﻿
CREATE PROCEDURE [dbo].[CatalogUpdate]
(
	@ParentId int,
	@Name varchar(100),
	@Note varchar(250) ,
	@MarkUp float,
	@MD datetime ,
	@MW varchar(50),
	@ToDelete bit,
	@Original_Id int,
	@Original_MD datetime
)
AS
DECLARE @strMsg nvarchar(500), @intRet int, @intErrCode int
DECLARE @orgToDelete bit, @orgParentToDelete bit
SET NOCOUNT ON

IF (@Name Is Null) OR (@ParentId Is Null) 
OR (@Original_Id Is Null) OR (@Original_MD Is Null) OR (@ToDelete Is Null)
BEGIN
	SELECT @intErrCode = 1, @strMsg = 'Неправильные параметры в вызове процедуры ' + Object_Name(@@Procid) + ' !'
	RAISERROR (@strMsg, 11, @intErrCode)
	RETURN @intErrCode
END

IF (ltrim(@Name) = '')
BEGIN
	SELECT @intErrCode = 2, @strMsg = 'Незаполнено наименование!'
	RAISERROR (@strMsg, 11, @intErrCode)
	RETURN @intErrCode
END

SELECT @orgToDelete = ToDelete
FROM [CO].[Catalog] WHERE Id = @Original_Id AND MD = @Original_MD
IF (@@rowcount <> 1)
BEGIN
	SELECT @intErrCode = 3, 
	@strMsg = 'Данные изменены после последнего запроса!'
	RAISERROR (@strMsg, 11, 1)
	RETURN @intErrCode
END

SELECT @orgParentToDelete = ToDelete
FROM [CO].[Catalog] WHERE Id = @ParentId
IF (@@rowcount <> 1)
BEGIN
	SELECT @intErrCode = 4, 
	@strMsg = 'Отсутствует родительская запись с кодом ' + cast(@ParentId as varchar(20)) + ' !'
	RAISERROR (@strMsg, 11, 1)
	RETURN @intErrCode
END

IF @orgParentToDelete = 1 
BEGIN
	SELECT @intErrCode = 5, 
	@strMsg = 'Родительская запись с кодом ' + cast(@ParentId as varchar(20)) + ' отмечена для удаления!'
	RAISERROR (@strMsg, 11, 1)
	RETURN @intErrCode
END

IF @orgToDelete = 1 --Запись была отмечена для удаления
BEGIN
	IF @ToDelete = 1 
	BEGIN
		SELECT @intErrCode = 6, 
		@strMsg = 'Нельзя редактировать запись, отмеченную для удаления!'
		RAISERROR (@strMsg, 11, 1)
		RETURN @intErrCode
	END
END ELSE BEGIN
	IF @ToDelete = 1 
	BEGIN
		SELECT @intErrCode = 7, 
		@strMsg = 'Для удаления данных должна использоваться другая процедура!'
		RAISERROR (@strMsg, 11, 1)
		RETURN @intErrCode
	END
END

EXEC [dbo].[ProcessStatistic] @Note OUTPUT, @MD OUTPUT, @MW OUTPUT
	SET NOCOUNT OFF;
UPDATE [CO].[Catalog] SET [ParentId] = @ParentId, [Name] = @Name, MarkUp = @MarkUp,
[Note] = @Note, [MD] = @MD, [MW] = @MW, ToDelete = @ToDelete 
WHERE Id = @Original_Id AND MD = @Original_MD;

SELECT ParentId, Name, MarkUp, Note, MD, MW, ToDelete FROM CO.Catalog WHERE (Id = @Original_Id)

RETURN 0
